

## Workshop 2

```
DEFINE DATA LOCAL                                /* IW2DEMO      (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 MIDDLE-NAME
  2 SEX
END-DEFINE
*
READ (15) EMP BY PERSONNEL-ID
  DISPLAY EMP
END-READ
END
```

## Workshop 3

```
DEFINE DATA LOCAL                                /* IW3VAR      (c) LEN Consulting LLC
1 #ALPHA (A256)                                  /* Y: 253 pre-Nat4
1 #BINARY (B200)                                /* Y: 128 pre-Nat4
1 #GIANT (A2000000000)                          /* N: 1G maximum
1 #A2 (I3)                                       /* N: length 1, 2, 4
1 #DATE (D6)                                     /* N: implied length 6
1 #HIGH-VALUES (H1)                             /* N: invalid format
1 VERY_DESCRIPTIVE_FIELD_NAME (A1)             /* Y: supports DB2/SQL
1 $AMOUNT (P13.2)                               /* N: Cannot begin with $
1 #AMOUNT$ (P13.2)                             /* Y
1 #TOTAL/2 (P7.4)                              /* Y: is not arithmetic
1 READ (I4)                                     /* Y: but a Natural keyword
1 #GLOBALS (P27.4)                             /* N: Too many digits
END-DEFINE
END
```

```
DEFINE DATA LOCAL                                /* IW3NUM          (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 LEAVE-DUE
  2 LEAVE-TAKEN
  2 SEX
END-DEFINE
*
READ (15) EMP BY PERSONNEL-ID
  DISPLAY (ZP=F)
  EMP
END-READ
END
```

Workshop 4

```
DEFINE DATA LOCAL                                /* IW4DATE      (c) LEN Consulting LLC
1 #DATE (D)   INIT <D'06/27/2007'>
1 #MON (D)    INIT <D'07/01/2007'>
1 #TIME (T)   INIT <D'06/27/2007'>
1 #DAYS (N3)
END-DEFINE
*
WRITE '=' #DATE
WRITE '=' #DATE (EM=MM/DD/YYYY^LLLLLLLLLLL^^NNNNNNNNNN)
WRITE '=' #MON  (EM=MM/DD/YYYY)
ASSIGN #DAYS = #MON - #DATE
WRITE '=' #DAYS
ADD 120 TO #DATE
WRITE '=' #DATE (EM=MM/DD/YYYY^LLLLLLLLLLL^^NNNNNNNNNN)
END
```

```
DEFINE DATA LOCAL                                /* IW4INPUT      (c) LEN Consulting LLC
1 #LEFT (P7.3)
1 #RIGHT (P7.3)
1 #OP (A1)          INIT <'+'>
1 #RESULT (P7.2)
1 #ROUND (A1)
END-DEFINE
*
REPEAT
  INPUT (AD=MDLT'_' IP=F)
      *PROGRAM (AD=O)
      10T 'Adding Machine'
      // 10T 'Left      Op      Right = Result'
      // 10T #LEFT
      #OP
      #RIGHT
      ' =' #RESULT (AD=O)
      #ROUND '(rounded)'
END-REPEAT
END
```

```

DEFINE DATA LOCAL                                /* IW4CALCI      (c) LEN Consulting LLC
1 #LEFT (P7.3)
1 #RIGHT (P7.3)
1 #OP (A1)          INIT <'+'>
1 #RESULT (P7.2)
1 #ROUND (A1)
END-DEFINE
*
REPEAT
  INPUT (AD=MDLT'_' IP=F)
          *PROGRAM (AD=0)
          10T 'Adding Machine - Nested IF'
          // 10T 'Left      Op      Right = Result'
          // 10T #LEFT
              #OP
              #RIGHT
              ' =' #RESULT (AD=OL)
              #ROUND '(rounded)'
IF #OP = '.'
  THEN
  STOP
END-IF
IF #OP = '+'
  AND #ROUND <> ' '
  THEN
  ADD ROUNDED #LEFT #RIGHT GIVING #RESULT
ELSE
  IF #OP = '+'
  THEN
  ADD #LEFT #RIGHT GIVING #RESULT
ELSE
  IF #OP = '-'
  AND #ROUND <> ' '
  THEN
  SUBTRACT ROUNDED #RIGHT FROM #LEFT GIVING #RESULT
ELSE
  IF #OP = '-'
  THEN
  SUBTRACT #RIGHT FROM #LEFT GIVING #RESULT
ELSE
  IF #OP = 'x'
  OR = '*'
  AND #ROUND <> ' '
  THEN
  MULTIPLY ROUNDED #LEFT BY #RIGHT GIVING #RESULT
ELSE
  IF #OP = 'x'
  OR = '*'
  THEN
  MULTIPLY #LEFT BY #RIGHT GIVING #RESULT
ELSE
  IF #OP = '/'
  AND #ROUND <> ' '
  THEN
  DIVIDE ROUNDED #RIGHT INTO #LEFT
  GIVING #RESULT
ELSE
  IF #OP = '/'
  THEN
  DIVIDE #RIGHT INTO #LEFT

```

```

                                GIVING #RESULT
                                ELSE
                                RESET #RESULT
                                END-IF
                                END-IF
                                END-IF
                                END-IF
                                END-IF
                                END-IF
                                END-IF
                                END-IF
                                END-REPEAT
                                END
```

```

DEFINE DATA LOCAL                                /* IW4CALCO      (c) LEN Consulting LLC
1 #LEFT (P7.3)
1 #RIGHT (P7.3)
1 #OP (A1)          INIT <'+'>
1 #RESULT (P7.2)
1 #ROUND (A1)
END-DEFINE
*
REPEAT
  INPUT (AD=MDLT'_' IP=F)
      *PROGRAM (AD=0)
      10T 'Adding Machine - DECIDE ON'
      // 10T 'Left      Op      Right = Result'
      // 10T #LEFT
          #OP
          #RIGHT
          ' =' #RESULT (AD=OL)
          #ROUND '(rounded)'
  IF #OP = '.'
      THEN
          STOP
  END-IF
  DECIDE ON FIRST #OP
  VALUE '+'
      IF #ROUND <> ' '
          THEN
              ADD ROUNDED #LEFT #RIGHT GIVING #RESULT
          ELSE
              ADD #LEFT #RIGHT GIVING #RESULT
          END-IF
  VALUE '-'
      IF #ROUND <> ' '
          THEN
              SUBTRACT ROUNDED #RIGHT FROM #LEFT GIVING #RESULT
          ELSE
              SUBTRACT #RIGHT FROM #LEFT GIVING #RESULT
          END-IF
  VALUE 'x', '*'
      IF #ROUND <> ' '
          THEN
              MULTIPLY ROUNDED #LEFT BY #RIGHT GIVING #RESULT
          ELSE
              MULTIPLY #LEFT BY #RIGHT GIVING #RESULT
          END-IF
  VALUE '/'
      IF #ROUND <> ' '
          THEN
              DIVIDE ROUNDED #RIGHT INTO #LEFT GIVING #RESULT
          ELSE
              DIVIDE #RIGHT INTO #LEFT GIVING #RESULT
          END-IF
  NONE
      RESET #RESULT
  END-DECIDE
END-REPEAT
END

```



```

DEFINE DATA LOCAL                                /* IW4CALCF      (c) LEN Consulting LLC
1 #LEFT (P7.3)
1 #RIGHT (P7.3)
1 #OP (A1)          INIT <'+'>
1 #RESULT (P7.2)
1 #ROUND (A1)
END-DEFINE
*
REPEAT
  INPUT (AD=MDLT'_' IP=F)
      *PROGRAM (AD=0)
      10T 'Adding Machine - DECIDE FOR'
      // 10T 'Left      Op      Right = Result'
      // 10T #LEFT
          #OP
          #RIGHT
          ' =' #RESULT (AD=OL)
          #ROUND '(rounded)'
  IF #OP = '.'
      THEN
          STOP
  END-IF
  DECIDE FOR FIRST CONDITION
  WHEN #OP = '+'
      AND #ROUND <> ' '
      ADD ROUNDED #LEFT #RIGHT GIVING #RESULT
  WHEN #OP = '+'
      ADD #LEFT #RIGHT GIVING #RESULT
  WHEN #OP = '-'
      AND #ROUND <> ' '
      SUBTRACT ROUNDED #RIGHT FROM #LEFT GIVING #RESULT
  WHEN #OP = '-'
      SUBTRACT #RIGHT FROM #LEFT GIVING #RESULT
  WHEN #OP = 'x'
      OR = '*'
      AND #ROUND <> ' '
      MULTIPLY ROUNDED #LEFT BY #RIGHT GIVING #RESULT
  WHEN #OP = 'x'
      OR = '*'
      MULTIPLY #LEFT BY #RIGHT GIVING #RESULT
  WHEN #OP = '/'
      AND #ROUND <> ' '
      DIVIDE ROUNDED #RIGHT INTO #LEFT GIVING #RESULT
  WHEN #OP = '/'
      DIVIDE #RIGHT INTO #LEFT GIVING #RESULT
  WHEN NONE
      RESET #RESULT
  END-DECIDE
END-REPEAT
END

```

## Workshop 5

```

DEFINE DATA LOCAL                                /* IW5RPT          (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 SEX
  2 LEAVE-DUE
END-DEFINE
*
READ (100) EMP BY NAME
  WRITE
      EMP.NAME (IS=T CD=BL)
      EMP.FIRST-NAME (AL=15)
      2X EMP.SEX
      3X EMP.LEAVE-DUE (EM=Z9)
      3X EMP.PERSONNEL-ID
END-READ
*
WRITE TITLE LEFT
      *PROGRAM
      10T 'Introduction to Natural' (YE)
      64T *DAT4U
      *TIME (AL=5)
/      *LIBRARY-ID
      10T 'Workshop 5 - Output' (RE)
      68T 'Page:' *PAGE-NUMBER (EM=ZZ,ZZ9)
//      'Surname'
      22T 'First Name'
      38T 'Sex'
      2X 'LD'
      6X 'ID'
/
END

```

- Methods of setting a field's header for a DISPLAY statement.
- Three methods to limit a database access loop to a specific number of iterations

## Workshop 6

```
DEFINE DATA LOCAL                                /* IW6PGM          (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 DEPT
  2 PERSONNEL-ID
  2 NAME
  2 LEAVE-DUE
*
1 #DEPT (A6)  INIT <'COMP02'>
END-DEFINE
*
REPEAT
  INPUT USING MAP 'IW6MAP'
  IF #DEPT = '.'
    THEN
      STOP
    END-IF
  READ EMP BY DEPT FROM #DEPT
  IF EMP.DEPT > #DEPT
    THEN
      ESCAPE BOTTOM
    END-IF
  DISPLAY EMP.DEPT
           EMP.NAME
           EMP.PERSONNEL-ID
           EMP.LEAVE-DUE
  END-READ
  RESET *PAGE-NUMBER
END-REPEAT
END
```

## Workshop 7

```
DEFINE DATA PARAMETER                /* IW7SUBP      (c) LEN Consulting LLC
1 #PID (A8)
1 #FIRST (A20)
1 #NAME (A20)
1 #AREA (A6)
1 #PHONE (A15)
LOCAL
1 EMP      VIEW EMPLOYEES
  2 FIRST-NAME
  2 NAME
  2 AREA-CODE
  2 PHONE
END-DEFINE
*
FIND EMP WITH PERSONNEL-ID = #PID
  ASSIGN #FIRST = EMP.FIRST-NAME
  ASSIGN #NAME  = EMP.NAME
  ASSIGN #AREA  = EMP.AREA-CODE
  ASSIGN #PHONE = EMP.PHONE
END-FIND
END
```

```
DEFINE DATA PARAMETER                /* IW7SUBPM     (c) LEN Consulting LLC
1 #EMP
  2 PERSONNEL-ID (A8)
  2 FIRST-NAME (A20)
  2 NAME (A20)
  2 AREA-CODE (A6)
  2 PHONE (A15)
LOCAL
1 EMP      VIEW EMPLOYEES
  2 FIRST-NAME
  2 NAME
  2 AREA-CODE
  2 PHONE
END-DEFINE
*
FIND EMP WITH PERSONNEL-ID = #EMP.PERSONNEL-ID
  MOVE BY NAME EMP TO #EMP
END-FIND
END
```

```
DEFINE DATA LOCAL                                /* IW7MAIN      (c) LEN Consulting LLC
1 #PID (A8)          INIT <30000231>
1 #FIRST (A20)
1 #NAME (A20)
1 #AREA (A6)
1 #PHONE (A15)
END-DEFINE
*
REPEAT
  INPUT #PID (AD=MDLT'_' )
  CALLNAT 'IW7SUBP' #PID
*   CALLNAT 'IW7SUBPM' #PID
                                #FIRST
                                #NAME
                                #AREA
                                #PHONE
  PERFORM DETAIL-REPORT
END-REPEAT
*
DEFINE SUBROUTINE DETAIL-REPORT
DISPLAY #PID
      #FIRST
      #NAME
      #AREA
      #PHONE
END-SUBROUTINE
END
```

```

DEFINE DATA LOCAL                                /* IW7MAINR      (c) LEN Consulting LLC
1 #PID (A8)          INIT <30000231>
1 #FIRST (A20)
1 #NAME (A20)
1 #AREA (A6)
1 #PHONE (A15)
END-DEFINE
*
REPEAT
  INPUT #PID (AD=MDLT'_' )
  PERFORM EMP-ROUTINE #PID
                        #FIRST
                        #NAME
                        #AREA
                        #PHONE
  PERFORM DETAIL-REPORT
END-REPEAT
*
DEFINE SUBROUTINE DETAIL-REPORT
DISPLAY #PID
        #FIRST
        #NAME
        #AREA
        #PHONE
END-SUBROUTINE
END

```

```

DEFINE DATA PARAMETER                            /* IW7SUBR      (c) LEN Consulting LLC
1 #PID (A8)
1 #FIRST (A20)
1 #NAME (A20)
1 #AREA (A6)
1 #PHONE (A15)
LOCAL
1 EMP      VIEW EMPLOYEES
  2 FIRST-NAME
  2 NAME
  2 AREA-CODE
  2 PHONE
END-DEFINE
*
DEFINE SUBROUTINE EMP-ROUTINE
FIND EMP WITH PERSONNEL-ID = #PID
  ASSIGN #FIRST = EMP.FIRST-NAME
  ASSIGN #NAME  = EMP.NAME
  ASSIGN #AREA  = EMP.AREA-CODE
  ASSIGN #PHONE = EMP.PHONE
END-FIND
END-SUBROUTINE
END

```

Workshop 8

```
DEFINE DATA LOCAL                                /* IW8LDA1      (c) LEN Consulting LLC
  1 EMP VIEW OF EMPLOYEES
    2 DEPT (A6)
    2 PERSONNEL-ID (A8)
    2 NAME (A20)
    2 LEAVE-DUE (N2.0)
  1 #DEPT (A6)
    INIT
    <'COMP02'>
  1 #RALPH (A10) INIT
    <'Zbrog'>
  1 #ZBROG (I4)
END-DEFINE
```

```
DEFINE DATA LOCAL                                /* IW8LDA2      (c) LEN Consulting LLC
  1 EMP VIEW OF EMPLOYEES
    2 PERSONNEL-ID (A8)
    2 FIRST-NAME (A20)
    2 NAME (A20)
    2 DEPT (A6)
    2 JOB-TITLE (A25)
    2 LEAVE-DUE (N2.0)
    2 LEAVE-TAKEN (N2.0)
END-DEFINE
```

## Workshop 9

```

DEFINE DATA LOCAL                                /* IW9NEST      (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 SEX
1 VEH      VIEW VEHICLES
  2 MAKE
  2 MODEL
  2 COLOR
  2 YEAR
END-DEFINE
*
FORMAT LS=132 PS=55
*
READ (100) EMP BY NAME
  FIND VEH WITH PERSONNEL-ID = EMP.PERSONNEL-ID
  WRITE
      EMP.NAME (IS=T CD=BL)
      EMP.FIRST-NAME (AL=15)
      2X EMP.SEX
      3X EMP.PERSONNEL-ID
      VEH.MAKE
      VEH.MODEL
      VEH.COLOR
      VEH.YEAR

  END-FIND
END-READ
*
WRITE TITLE LEFT
  *PROGRAM
  10T 'Introduction to Natural' (YE)
  114T *DAT4U
  *TIME (AL=5)
  / *LIBRARY-ID
  10T 'Workshop 9' (RE)
  118T 'Page:' *PAGE-NUMBER (EM=ZZ,ZZ9)
  // 'Surname'
  22T 'First Name'
  38T 'Sex'
  2X 'ID'
  7X 'Make'
  17X 'Model'
  16X 'Color'
  7X 'Year'
  /
END

```



```

DEFINE DATA LOCAL                                /* IW9HIST      (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 DEPT
END-DEFINE
*
HISTOGRAM EMP FOR DEPT
  DISPLAY EMP.DEPT
          *NUMBER
END-HISTOGRAM
END

```

```

DEFINE DATA LOCAL                                /* IW9HIST2     (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 DEPT-PERSON
END-DEFINE
*
HISTOGRAM (15) EMP FOR DEPT-PERSON
  DISPLAY EMP.DEPT-PERSON (EM=XXXX^XX^X(20))
          *NUMBER
END-HISTOGRAM
END

```

```

DEFINE DATA LOCAL                                /* IW9HIST3     (c) LEN Consulting LLC
1 EMP      VIEW EMPLOYEES
  2 DEPT-PERSON      2 REDEFINE DEPT-PERSON
    3 #DEPT (A4)
    3 #DIV (N2)
    3 #PERSON (A20)
END-DEFINE
*
HISTOGRAM (15) EMP FOR DEPT-PERSON
  DISPLAY #DEPT
          #DIV
          #PERSON
          *NUMBER
END-HISTOGRAM
END

```